Сочинил отчёт по оплатам менеджеров. Всё устраивает, кроме одного: даты получаются несколькими строками, то есть все три временные таблицы добавляют записи, а надо чтобы дни "схлопывались".
Может, сможете помочь? Где косяк?

```1c
    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдер.Дата, ДЕНЬ) КАК Дата,
    ПриходныйКассовыйОрдер.ДокументОснование.Менеджер КАК Менеджер,
    СУММА(0) КАК Заказано,
    СУММА(ПриходныйКассовыйОрдер.СуммаДокумента) КАК СуммаНал,
    СУММА(0) КАК СуммаБезнал
    ИЗ
    Документ.ПриходныйКассовыйОрдер КАК ПриходныйКассовыйОрдер
    ГДЕ
    ПриходныйКассовыйОрдер.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПриходныйКассовыйОрдер.Проведен = ИСТИНА
    И ВЫБОР
    КОГДА ПриходныйКассовыйОрдер.ДокументОснование = НЕОПРЕДЕЛЕНО
    ТОГДА ЛОЖЬ
    ИНАЧЕ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.ЗаказКлиента
    ИЛИ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.РеализацияТоваровУслуг
    КОНЕЦ
    И ПриходныйКассовыйОрдер.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдер.Дата, ДЕНЬ),
    ПриходныйКассовыйОрдер.ДокументОснование.Менеджер

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата, ДЕНЬ),
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.ОснованиеПлатежа.Менеджер,
    СУММА(0),
    СУММА(0),
    СУММА(ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Сумма)
    ИЗ
    Документ.ПоступлениеБезналичныхДенежныхСредств.РасшифровкаПлатежа КАК ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа
    ГДЕ
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Проведен = ИСТИНА
    И (ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиентаПоПлатежнойКарте)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыПоПлатежнойКарте))

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата, ДЕНЬ),
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.ОснованиеПлатежа.Менеджер

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ЗаказКлиента.Дата, ДЕНЬ),
    ЗаказКлиента.Менеджер,
    СУММА(ЗаказКлиента.СуммаДокумента),
    СУММА(0),
    СУММА(0)
    ИЗ
    Документ.ЗаказКлиента КАК ЗаказКлиента
    ГДЕ
    ЗаказКлиента.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ЗаказКлиента.Проведен = ИСТИНА

    СГРУППИРОВАТЬ ПО
    НАЧАЛОПЕРИОДА(ЗаказКлиента.Дата, ДЕНЬ),
    ЗаказКлиента.Менеджер

    УПОРЯДОЧИТЬ ПО
    Менеджер,
    Дата
```
---
```1c
    ВЫБРАТЬ
    ВложенныйЗапрос.Дата КАК Дата,
    ВложенныйЗапрос.Менеджер КАК Менеджер,
    СУММА(ВложенныйЗапрос.Заказано) КАК Заказано,
    СУММА(ВложенныйЗапрос.СуммаНал) КАК СуммаНал,
    СУММА(ВложенныйЗапрос.СуммаБезнал) КАК СуммаБезнал

    ИЗ
    (ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдер.Дата, ДЕНЬ) КАК Дата,
    ПриходныйКассовыйОрдер.ДокументОснование.Менеджер КАК Менеджер,
    0 КАК Заказано,
    ПриходныйКассовыйОрдер.СуммаДокумента КАК СуммаНал,
    0 КАК СуммаБезнал
    ИЗ
    Документ.ПриходныйКассовыйОрдер КАК ПриходныйКассовыйОрдер
    ГДЕ
    ПриходныйКассовыйОрдер.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПриходныйКассовыйОрдер.Проведен = ИСТИНА
    И ВЫБОР
    КОГДА ПриходныйКассовыйОрдер.ДокументОснование = НЕОПРЕДЕЛЕНО
    ТОГДА ЛОЖЬ
    ИНАЧЕ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.ЗаказКлиента
    ИЛИ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.РеализацияТоваровУслуг
    КОНЕЦ
    И ПриходныйКассовыйОрдер.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата, ДЕНЬ),
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.ОснованиеПлатежа.Менеджер,
    0,
    0,
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Сумма
    ИЗ
    Документ.ПоступлениеБезналичныхДенежныхСредств.РасшифровкаПлатежа КАК ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа
    ГДЕ
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Проведен = ИСТИНА
    И (ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиентаПоПлатежнойКарте)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыПоПлатежнойКарте))

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ЗаказКлиента.Дата, ДЕНЬ),
    ЗаказКлиента.Менеджер,
    ЗаказКлиента.СуммаДокумента,
    0,
    0
    ИЗ
    Документ.ЗаказКлиента КАК ЗаказКлиента
    ГДЕ
    ЗаказКлиента.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ЗаказКлиента.Проведен = ИСТИНА) КАК ВложенныйЗапрос

    СГРУППИРОВАТЬ ПО
    ВложенныйЗапрос.Дата,
    ВложенныйЗапрос.Менеджер

    УПОРЯДОЧИТЬ ПО
    ВложенныйЗапрос.Менеджер,
    ВложенныйЗапрос.Дата
```
---
Поместить результат объединения во временную таблицу и сгруппировать по дате
---
```1c
    ВЫБРАТЬ
    ВложенныйЗапрос.Дата КАК Дата,
    ВложенныйЗапрос.Менеджер КАК Менеджер,
    СУММА(ВложенныйЗапрос.Заказано) КАК Заказано,
    СУММА(ВложенныйЗапрос.СуммаНал) КАК СуммаНал,
    СУММА(ВложенныйЗапрос.СуммаБезнал) КАК СуммаБезнал

    ИЗ
    (ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПриходныйКассовыйОрдер.Дата, ДЕНЬ) КАК Дата,
    ПриходныйКассовыйОрдер.ДокументОснование.Менеджер КАК Менеджер,
    0 КАК Заказано,
    ПриходныйКассовыйОрдер.СуммаДокумента КАК СуммаНал,
    0 КАК СуммаБезнал
    ИЗ
    Документ.ПриходныйКассовыйОрдер КАК ПриходныйКассовыйОрдер
    ГДЕ
    ПриходныйКассовыйОрдер.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПриходныйКассовыйОрдер.Проведен = ИСТИНА
    И ВЫБОР
    КОГДА ПриходныйКассовыйОрдер.ДокументОснование = НЕОПРЕДЕЛЕНО
    ТОГДА ЛОЖЬ
    ИНАЧЕ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.ЗаказКлиента
    ИЛИ ПриходныйКассовыйОрдер.ДокументОснование ССЫЛКА Документ.РеализацияТоваровУслуг
    КОНЕЦ
    И ПриходныйКассовыйОрдер.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата, ДЕНЬ),
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.ОснованиеПлатежа.Менеджер,
    0,
    0,
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Сумма
    ИЗ
    Документ.ПоступлениеБезналичныхДенежныхСредств.РасшифровкаПлатежа КАК ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа
    ГДЕ
    ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.Проведен = ИСТИНА
    И (ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиента)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыОтКлиентаПоПлатежнойКарте)
    ИЛИ ПоступлениеБезналичныхДенежныхСредствРасшифровкаПлатежа.Ссылка.ХозяйственнаяОперация = ЗНАЧЕНИЕ(Перечисление.ХозяйственныеОперации.ПоступлениеОплатыПоПлатежнойКарте))

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    НАЧАЛОПЕРИОДА(ЗаказКлиента.Дата, ДЕНЬ),
    ЗаказКлиента.Менеджер,
    ЗаказКлиента.СуммаДокумента,
    0,
    0
    ИЗ
    Документ.ЗаказКлиента КАК ЗаказКлиента
    ГДЕ
    ЗаказКлиента.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания
    И ЗаказКлиента.Проведен = ИСТИНА) КАК ВложенныйЗапрос

    СГРУППИРОВАТЬ ПО
    ВложенныйЗапрос.Дата,
    ВложенныйЗапрос.Менеджер

    УПОРЯДОЧИТЬ ПО
    ВложенныйЗапрос.Менеджер,
    ВложенныйЗапрос.Дата
```
---
SlavaKron - спасибо, добрый человек, Век живи, век учись...